



with the turtle facing it. Now set up a WHEN demon with the following command:

```
WHEN 0 [BK 50]
```

Nothing happens immediately, but the WHEN demon is now present inside the computer keeping a watch out for event 0. Now try SETSP 30. The turtle sets off towards the line, but when it reaches it the WHEN demon is triggered and the turtle is thrown back. The turtle keeps on going with a speed of 30, but every time it comes to the line the WHEN demon repulses it.

This WHEN demon will stay in operation until you clear it by typing WHEN 0 []. All demons will be removed if you type CS, if there is an error message, or if you use the editor.

It is a nuisance to have to remember all the codes for the various collisions, so there are two primitives to help you: OVER <turtle number> <pen number> outputs the number for the collision between that turtle and a line drawn in that pen, and TOUCHING <turtle number 1> <turtle number 2> outputs the number of the demon for a collision between those two turtles.

CAGING TURTLE

Here is a set of procedures to cage a turtle within a box. Whenever the turtle hits the edge of the box (WHEN OVER 0 0) a demon calls up the procedure TURN. This causes the turtle to retreat 10 units, and then make a random turn. (RANDOM, together with a number, N, outputs a random number between 0 and N-1 inclusive).

```
TO TRAP
  DRAW TRAP
  HOME
  WHEN OVER 0 0 [TURN]
  SETSP 50
END
```

```
TO DRAW TRAP
  CS
  PU
  SETPOS [-50 -50]
  PD
  SQUARE
  PU
END
```

```
TO SQUARE
  REPEAT 4 [FD 100 RT 90]
END
```

```
TO TURN
  BK 10
  RT RANDOM 45
END
```

Demons can also be used for watching the

joystick. Of the 21 special events that we have mentioned, event 3 occurs when the joystick button is pressed, and event 15 occurs when the joystick position is changed. The command JOY 1 outputs a number from -1 to 7 corresponding to the position of the joystick (in port 2). Define JOYH in this way:

```
TO JOYH
  IF (JOY 1) < 0 [STOP]
  ASK 0 [SETH 45 * JOY 1]
END
```

then set the turtle in motion with SETSP 50, and finally set up a WHEN demon:

```
WHEN 15 [JOYH]
```

The joysticks can now be used to control the heading of turtle 0.

You can give more than one WHEN command at the same time, but they are not actually active simultaneously. While one demon is busy (i.e. when its event occurs) the others are inactive. This can mean that some collisions go undetected.

The way to deal with this problem is to have each demon simply set the speeds to zero and to run a continuous procedure that watches for this happening. Adapting our previous program to use this technique:

```
TO TRAP
  DRAW TRAP
  HOME
  WHEN OVER 0 0 [SETSP 0]
  SETSP 50
  WATCH
END

TO WATCH
  IF :SPEED = 0 [CHECK]
  WATCH
END
```

In this procedure SPEED outputs the value of the speed of the current turtle. The procedure CHECK must determine which event has occurred, carry out the necessary actions and then restore the speeds. In this case there is only one event we are interested in, but it illustrates the way to program the method.

```
TO CHECK
  IF COND OVER 0 0 THEN [TURN]
  SETSP 50
END
```

The command COND and a number gives a true output if an event of that number has occurred. COND can only check an event at the time LOGO executes the line containing it.

